# **NAG Library Routine Document**

## H05AAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of **bold italicised** terms and other implementation-dependent details.

## 1 Purpose

Given a set of m features and a scoring mechanism for any subset of those features, H05AAF selects the best n subsets of size p using a reverse communication branch and bound algorithm.

## 2 Specification

```
SUBROUTINE H05AAF (IREVCM, MINCR, M, IP, NBEST, DROP, LZ, Z, LA, A, BSCORE, BZ, MINCNT, GAMMA, ACC, ICOMM, LICOMM, RCOMM, LRCOMM, IFAIL)

INTEGER IREVCM, MINCR, M, IP, NBEST, DROP, LZ, Z(M-IP), LA, A(max(NBEST,M)), BZ(M-IP,NBEST), MINCNT, ICOMM(LICOMM), LICOMM, LRCOMM, IFAIL

REAL (KIND=nag_wp) BSCORE(max(NBEST,M)), GAMMA, ACC(2), RCOMM(LRCOMM)
```

## 3 Description

Given  $\Omega = \{x_i : i \in \mathbb{Z}, 1 \le i \le m\}$ , a set of m unique features and a scoring mechanism f(S) defined for all  $S \subseteq \Omega$  then H05AAF is designed to find  $S_{o1} \subseteq \Omega, |S_{o1}| = p$ , an optimal subset of size p. Here  $|S_{o1}|$  denotes the cardinality of  $S_{o1}$ , the number of elements in the set.

The definition of the optimal subset depends on the properties of the scoring mechanism, if

$$f(S_i) \le f(S_j)$$
, for all  $S_i \subseteq \Omega$  and  $S_i \subseteq S_j$  (1)

then the optimal subset is defined as one of the solutions to

$$\underset{S\subseteq\varOmega}{\operatorname{maximize}}f\left(S\right)\qquad\text{subject to}\qquad\left|S\right|=p$$

else if

$$f(S_i) \ge f(S_j)$$
, for all  $S_j \subseteq \Omega$  and  $S_i \subseteq S_j$  (2)

then the optimal subset is defined as one of the solutions to

$$\underset{S\subseteq\varOmega}{\operatorname{minimize}}f\left(S\right)\qquad\text{subject to}\qquad\left|S\right|=p$$

If neither of these properties hold then H05AAF cannot be used.

As well as returning the optimal subset,  $S_{o1}$ , H05AAF can return the best n solutions of size p. If  $S_{oi}$  denotes the ith best subset, for i = 1, 2, ..., n-1, then the (i+1)th best subset is defined as the solution to either

or

$$\underset{S\subseteq \varOmega -\left\{ S_{oj}; j\in \mathbb{Z}, 1\leq j\leq i\right\} }{\text{minimize}} f\left(S\right) \qquad \text{subject to} \qquad |S|=p$$

depending on the properties of f.

The solutions are found using a branch and bound method, where each node of the tree is a subset of  $\Omega$ . Assuming that (1) holds then a particular node, defined by subset  $S_i$ , can be trimmed from the tree if  $f(S_i) < \hat{f}(S_{on})$  where  $\hat{f}(S_{on})$  is the *n*th highest score we have observed so far for a subset of size p, i.e.,

our current best guess of the score for the *n*th best subset. In addition, because of (1) we can also drop all nodes defined by any subset  $S_j$  where  $S_j \subseteq S_i$ , thus avoiding the need to enumerate the whole tree. Similar short cuts can be taken if (2) holds. A full description of this branch and bound algorithm can be found in Ridout (1988).

Rather than calculate the score at a given node of the tree H05AAF utilizes the fast branch and bound algorithm of Somol *et al.* (2004), and attempts to estimate the score where possible. For each feature,  $x_i$ , two values are stored, a count  $c_i$  and  $\hat{\mu}_i$ , an estimate of the contribution of that feature. An initial value of zero is used for both  $c_i$  and  $\hat{\mu}_i$ . At any stage of the algorithm where both f(S) and  $f(S - \{x_i\})$  have been calculated (as opposed to estimated), the estimated contribution of the feature  $x_i$  is updated to

$$\frac{c_{i}\hat{\mu}_{i}+\left[f\left(S\right)-f\left(S-\left\{ x_{j}\right\} \right)\right]}{c_{i}+1}$$

and  $c_i$  is incremented by 1, therefore at each stage  $\hat{\mu}_i$  is the mean contribution of  $x_i$  observed so far and  $c_i$  is the number of observations used to calculate that mean.

As long as  $c_i \ge k$ , for the user-supplied constant k, then rather than calculating  $f(S - \{x_i\})$  this routine estimates it using  $\hat{f}(S - \{x_i\}) = f(S) - \gamma \hat{\mu}_i$  or  $\hat{f}(S) - \gamma \hat{\mu}_i$  if f(S) has been estimated, where  $\gamma$  is a user-supplied scaling factor. An estimated score is never used to trim a node or returned as the optimal score.

Setting k=0 in this routine will cause the algorithm to always calculate the scores, returning to the branch and bound algorithm of Ridout (1988). In most cases it is preferable to use the fast branch and bound algorithm, by setting k>0, unless the score function is iterative in nature, i.e., f(S) must have been calculated before  $f(S-\{x_i\})$  can be calculated.

#### 4 References

Narendra P M and Fukunaga K (1977) A branch and bound algorithm for feature subset selection *IEEE Transactions on Computers* **9** 917–922

Ridout M S (1988) Algorithm AS 233: An improved branch and bound algorithm for feature subset selection *Journal of the Royal Statistics Society, Series C (Applied Statistics) (Volume 37)* **1** 139–147

Somol P, Pudil P and Kittler J (2004) Fast branch and bound algorithms for optimal feature selection *IEEE Transactions on Pattern Analysis and Machine Intelligence (Volume 26)* **7** 900–912

#### 5 Parameters

**Note**: this routine uses **reverse communication**. Its use involves an initial entry, intermediate exits and reentries, and a final exit, as indicated by the parameter **IREVCM**. Between intermediate exits and reentries, all parameters other than **BSCORE** must remain unchanged.

#### 1: IREVCM – INTEGER

Input/Output

On initial entry: must be set to 0.

*On intermediate exit*: IREVCM = 1 and before re-entry the scores associated with LA subsets must be calculated and returned in BSCORE.

The LA subsets are constructed as follows:

$$DROP = 1$$

The *j*th subset is constructed by *dropping* the features specified in the first LZ elements of Z and the single feature given in A(j) from the full set of features,  $\Omega$ . The subset will therefore contain M - LZ - 1 features.

$$DROP = 0$$

The *j*th subset is constructed by *adding* the features specified in the first LZ elements of Z and the single feature specified in A(j) to the empty set,  $\emptyset$ . The subset will therefore contain LZ + 1 features.

H05AAF.2 Mark 24

H05AAF

In both cases the individual features are referenced by the integers 1 to M with 1 indicating the first feature, 2 the second, etc., for some arbitrary ordering of the features. The same ordering must be used in all calls to H05AAF.

If LA = 0, the score for a single subset should be returned. This subset is constructed by adding or removing only those features specified in the first LZ elements of Z.

If LZ = 0, this subset will either be  $\Omega$  or  $\emptyset$ .

The score associated with the jth subset must be returned in BSCORE(j).

On intermediate re-entry: IREVCM must remain unchanged.

On final exit: IREVCM = 0, and the algorithm has terminated.

Constraint: IREVCM = 0 or 1.

#### 2: MINCR - INTEGER

Input

On entry: flag indicating whether the scoring function f is increasing or decreasing.

MINCR = 1

 $f(S_i) \leq f(S_i)$ , i.e., the subsets with the largest score will be selected.

MINCR = 0

 $f(S_i) \ge f(S_i)$ , i.e., the subsets with the smallest score will be selected.

For all  $S_j \subseteq \Omega$  and  $S_i \subseteq S_j$ .

Constraint: MINCR = 0 or 1.

#### 3: M – INTEGER

Input

On entry: m, the number of features in the full feature set.

Constraint:  $M \ge 2$ .

#### 4: IP – INTEGER

Input

On entry: p, the number of features in the subset of interest.

*Constraint*:  $1 \le IP \le M$ .

#### 5: NBEST – INTEGER

Input

On entry: n, the maximum number of best subsets required. The actual number of subsets returned is given by LA on final exit. If on final exit LA  $\neq$  NBEST then IFAIL = 53 is returned.

*Constraint*: NBEST  $\geq 1$ .

#### 6: DROP – INTEGER

Input/Output

On initial entry: DROP need not be set.

On intermediate exit: flag indicating whether the intermediate subsets should be constructed by dropping features from the full set (DROP = 1) or adding features to the empty set (DROP = 0). See IREVCM for details.

On intermediate re-entry: DROP must remain unchanged.

On final exit: is undefined.

### 7: LZ – INTEGER

Input/Output

On initial entry: LZ need not be set.

On intermediate exit: the number of features stored in Z.

On intermediate re-entry: LZ must remain unchanged.

On final exit: is undefined.

### 8: Z(M - IP) - INTEGER array

Input/Output

On initial entry: Z need not be set.

On intermediate exit: Z(i), for i = 1, 2, ..., LZ, contains the list of features which, along with those specified in A, define the subsets whose score is required. See IREVCM for additional details.

On intermediate re-entry: Z must remain unchanged.

On final exit: is undefined.

#### 9: LA – INTEGER

Input/Output

On initial entry: LA need not be set.

On intermediate exit: if LA > 0, the number of subsets for which a score must be returned.

If LA = 0, the score for a single subset should be returned. See IREVCM for additional details.

On intermediate re-entry: LA must remain unchanged.

On final exit: the number of best subsets returned.

#### 10: A(max(NBEST, M)) - INTEGER array

Input/Output

On initial entry: A need not be set.

On intermediate exit: A(j), for j = 1, 2, ..., LA, contains the list of features which, along with those specified in Z, define the subsets whose score is required. See IREVCM for additional details.

On intermediate re-entry: A must remain unchanged.

On final exit: is undefined.

### 11: BSCORE(max(NBEST, M)) – REAL (KIND=nag\_wp) array

Input/Output

On initial entry: BSCORE need not be set.

On intermediate exit: is undefined.

On intermediate re-entry: BSCORE(j) must hold the score for the jth subset as described in IREVCM.

On final exit: holds the score for the LA best subsets returned in BZ.

#### 12: BZ(M - IP, NBEST) - INTEGER array

Input/Output

On initial entry: BZ need not be set.

On intermediate exit: is used for storage between calls to H05AAF.

On intermediate re-entry: BZ must remain unchanged.

On final exit: the jth best subset is constructed by dropping the features specified in BZ(i,j), for  $i=1,2,\ldots,M-IP$  and  $j=1,2,\ldots,LA$ , from the set of all features,  $\Omega$ . The score for the jth best subset is given in BSCORE(j).

### 13: MINCNT – INTEGER

Input

On entry: k, the minimum number of times the effect of each feature,  $x_i$ , must have been observed before  $f(S - \{x_i\})$  is estimated from f(S) as opposed to being calculated directly.

If k = 0 then  $f(S - \{x_i\})$  is never estimated. If MINCNT < 0 then k is set to 1.

## 14: GAMMA – REAL (KIND=nag\_wp)

Input

On entry:  $\gamma$ , the scaling factor used when estimating scores. If GAMMA < 0 then  $\gamma$  is set to 1.

#### 15: ACC(2) – REAL (KIND=nag wp) array

Input

On entry: a measure of the accuracy of the scoring function, f.

H05AAF.4 Mark 24

Letting  $a_i = \epsilon_1 |f(S_i)| + \epsilon_2$ , then when confirming whether the scoring function is strictly increasing or decreasing (as described in MINCR), or when assessing whether a node defined by subset  $S_i$  can be trimmed, then any values in the range  $f(S_i) \pm a_i$  are treated as being numerically equivalent.

If 
$$0 \le ACC(1) \le 1$$
 then  $\epsilon_1 = ACC(1)$ , otherwise  $\epsilon_1 = 0$ .

If 
$$ACC(2) \ge 0$$
 then  $\epsilon_2 = ACC(2)$ , otherwise  $\epsilon_2 = 0$ .

In most situations setting both  $\epsilon_1$  and  $\epsilon_2$  to zero should be sufficient. Using a nonzero value, when one is not required, can significantly increase the number of subsets that need to be evaluated.

#### 16: ICOMM(LICOMM) – INTEGER array

Communication Array

On initial entry: ICOMM need not be set.

On intermediate exit: is used for storage between calls to H05AAF.

On intermediate re-entry: ICOMM must remain unchanged.

On final exit: is not defined. The first two elements, ICOMM(1) and ICOMM(2) contain the minimum required value for LICOMM and LRCOMM respectively.

#### 17: LICOMM - INTEGER

Input

On entry: the length of the array ICOMM. If LICOMM is too small and LICOMM  $\geq 2$  then IFAIL = 172 is returned and the minimum value for LICOMM and LRCOMM are given by ICOMM(1) and ICOMM(2) respectively.

Constraints:

18:

```
if MINCNT = 0 LICOMM \ge 2 \times max(NBEST, M) + M(M+2) + (M+1) \times max(M-IP, 1) + 27; otherwise LICOMM \ge 2 \times max(NBEST, M) + M(M+3) + (2M+1) \times max(M-IP, 1) + 25.
```

## RCOMM(LRCOMM) - REAL (KIND=nag\_wp) array

Communication Array

On initial entry: RCOMM need not be set.

On intermediate exit: is used for storage between calls to H05AAF.

On intermediate re-entry: RCOMM must remain unchanged.

On final exit: is not defined.

#### 19: LRCOMM – INTEGER

Input

On entry: the length of the array RCOMM. If LRCOMM is too small and LICOMM  $\geq 2$  then IFAIL = 172 is returned and the minimum value for LICOMM and LRCOMM are given by ICOMM(1) and ICOMM(2) respectively.

Constraints:

```
if MINCNT = 0, LRCOMM \geq 9 + NBEST + M × max(M – IP, 1); otherwise LRCOMM \geq 8 + M + NBEST + M × max(M – IP, 1).
```

## 20: IFAIL - INTEGER

Input/Output

On initial entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL  $\neq 0$  on exit, the recommended value is -1. When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.

On final exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

```
IFAIL = 11
        On entry, IREVCM = \langle value \rangle.
        Constraint: IREVCM = 0 or 1.
IFAIL = 21
        On entry, MINCR = \langle value \rangle.
        Constraint: MINCR = 0 or 1.
IFAIL = 22
        MINCR has changed between calls.
        On intermediate entry, MINCR = \langle value \rangle.
        On initial entry, MINCR = \langle value \rangle.
IFAIL = 31
        On entry, M = \langle value \rangle.
        Constraint: M \ge 2.
IFAIL = 32
        M has changed between calls.
        On intermediate entry, M = \langle value \rangle.
        On initial entry, M = \langle value \rangle.
IFAIL = 41
        On entry, IP = \langle value \rangle and M = \langle value \rangle.
        Constraint: 1 \le IP \le M. On entry, IP = \langle value \rangle and M = \langle value \rangle.
        Constraint: 1 \le IP \le M.
IFAIL = 42
        IP has changed between calls.
        On intermediate entry, IP = \langle value \rangle.
        On initial entry, IP = \langle value \rangle.
IFAIL = 51
        On entry, NBEST = \langle value \rangle.
        Constraint: NBEST \geq 1.
IFAIL = 52
        NBEST has changed between calls.
        On intermediate entry, NBEST = \langle value \rangle.
        On initial entry, NBEST = \langle value \rangle.
IFAIL = 53
        On entry, NBEST = \langle value \rangle.
        But only \(\langle value \rangle \) best subsets could be calculated.
```

H05AAF.6 Mark 24

```
IFAIL = 61
       DROP has changed between calls.
       On intermediate entry, DROP = \langle value \rangle.
       On initial entry, DROP = \langle value \rangle.
IFAIL = 71
       LZ has changed between calls.
       On entry, LZ = \langle value \rangle.
       On previous exit, LZ = \langle value \rangle.
IFAIL = 91
       LA has changed between calls.
       On entry, LA = \langle value \rangle.
       On previous exit, LA = \langle value \rangle.
IFAIL = 111
       BSCORE(\langle value \rangle = \langle value \rangle, which is inconsistent with the score for the parent node. Score for
       the parent node is \langle value \rangle. BSCORE(\langle value \rangle = \langle value \rangle, which is inconsistent with the score for the
       parent node. Score for the parent node is \( \frac{value}{\} \).
IFAIL = 131
       MINCNT has changed between calls.
       On intermediate entry, MINCNT = \langle value \rangle.
       On initial entry, MINCNT = \langle value \rangle.
IFAIL = 141
       GAMMA has changed between calls.
       On intermediate entry, GAMMA = \langle value \rangle.
       On initial entry, GAMMA = \langle value \rangle.
IFAIL = 151
       ACC(1) has changed between calls.
       On intermediate entry, ACC(1) = \langle value \rangle.
       On initial entry, ACC(1) = \langle value \rangle.
IFAIL = 152
       ACC(2) has changed between calls.
       On intermediate entry, ACC(2) = \langle value \rangle.
       On initial entry, ACC(2) = \langle value \rangle.
IFAIL = 161
       ICOMM has been corrupted between calls. ICOMM has been corrupted between calls.
IFAIL = 171
       On entry, LICOMM = \langle value \rangle, LRCOMM = \langle value \rangle.
       Constraint: LICOMM \geq \langle value \rangle, LRCOMM \geq \langle value \rangle.
       ICOMM is too small to return the required array sizes.
IFAIL = 172
       On entry, LICOMM = \langle value \rangle, LRCOMM = \langle value \rangle.
       Constraint: LICOMM \geq \langle value \rangle, LRCOMM \geq \langle value \rangle.
       The minimum required values for LICOMM and LRCOMM are returned in ICOMM(1) and
```

ICOMM(2) respectively. On entry, LICOMM =  $\langle value \rangle$ , LRCOMM =  $\langle value \rangle$ .

```
Constraint: LICOMM \geq \langle value \rangle, LRCOMM \geq \langle value \rangle.
The minimum required values for LICOMM and LRCOMM are returned in ICOMM(1) and ICOMM(2) respectively.
```

```
IFAIL = 181
```

RCOMM has been corrupted between calls.

```
IFAIL = -999
```

Dynamic memory allocation failed.

### 7 Accuracy

The subsets returned by H05AAF are guaranteed to be optimal up to the accuracy of your calculated scores.

## **8** Further Comments

The maximum number of unique subsets of size p from a set of m features is  $N = \frac{m!}{(m-p)!p!}$ . The efficiency of the branch and bound algorithm implemented in H05AAF comes from evaluating subsets at internal nodes of the tree, that is subsets with more than p features, and where possible trimming branches of the tree based on the scores at these internal nodes as described in Narendra and Fukunaga (1977). Because of this it is possible, in some circumstances, for more than N subsets to be evaluated. This will tend to happen when most of the features have a similar effect on the subset score.

If multiple optimal subsets exist with the same score, and NBEST is too small to return them all, then the choice of which of these optimal subsets is returned is arbitrary.

## 9 Example

This example finds the three linear regression models, with five variables, that have the smallest residual sums of squares when fitted to a supplied dataset. The data used in this example was simulated.

#### 9.1 Program Text

```
HO5AAF Example Program Text
!
   Mark 24 Release. NAG Copyright 2012.
    Module h05aafe_mod
      .. Use Statements ..
1
     Use nag_library, Only: nag_wp
1
      .. Implicit None Statement ..
     Implicit None
1
      .. Parameters
     Integer, Parameter
                                            :: nin = 5. nout = 6
!
      .. Derived Type Definitions ..
                                            :: calc_subset_data
     Туре
        Integer
                                            :: n, ldq, ldx
        Real (Kind=nag_wp)
                                            :: tol
        Real (Kind=nag_wp), Allocatable
                                            :: x(:,:), y(:), b(:), cov(:),
                                               h(:), p(:), q(:,:), res(:),
                                               se(:), wk(:), wt(:)
        Integer, Allocatable
                                            :: isx(:)
        Character (1)
                                            :: mean, weight
     End Type calc_subset_data
    Contains
     Subroutine calc_subset_score(m,drop,lz,z,la,a,bscore,cs)
        Calculate the score associated with a particular set of feature subsets.
!
        M,DROP,LZ,Z,LA,A and BSCORE are all as described in the documentation of
!
        HO5AAF.
```

H05AAF.8 Mark 24

```
CS - Variable of type CALC_SUBSET_DATA that holds any additional data
!
             required by this routine
!
        This particular example finds the set, of a given size, of explanatory
        variables that best fit a response variable when a linear regression
1
        model is used. Therefore the feature set is the set of all the
        explanatory variables and the best set of features is defined as set of
1
        explanatory variables that gives the smallest residual sums of squares.
        See the documentation for GO2DAF for details on linear regression
1
!
       models.
!
        .. Use Statements ..
        Use nag_library, Only: g02daf
!
        .. Scalar Arguments ..
        Type (calc_subset_data), Intent (Inout) :: cs
        Integer, Intent (In)
                                              :: drop, la, lz, m
        .. Array Arguments .. Real (Kind=nag_wp), Intent (Out) :: bscore(:)
        Integer, Intent (In)
                                              :: a(:), z(:)
!
        .. Local Scalars ..
        Real (Kind=nag_wp)
                                              :: rss
        Integer
                                              :: i, idf, ifail, inv_drop, ip,
                                                 irank
        Logical
                                              :: svd
        Character (200)
                                              :: line
        .. Intrinsic Procedures ..
        Intrinsic
                                              :: abs, allocated, count, max
        .. Executable Statements ..
        Continue
        Allocate various arrays and read in the data if this is the first time
        this routine has been called
!
        If (.Not. allocated(cs%x)) Then
          Read in the number of observations for the data used in the linear
!
          regression skipping any headings or blank lines
            Read (nin, *, Iostat=ifail) line
            If (ifail/=0) Exit
            Read (line,*,Iostat=ifail) cs%n
            If (ifail==0) Exit
          End Do
          Read in the control parameters for the linear regression
          Read (nin,*) cs%tol
!
          Read in the data
          cs%ldx = cs%n
          Allocate (cs%x(cs%ldx,m),cs%y(cs%n))
          Read (nin,*)(cs%x(i,1:m),cs%y(i),i=1,cs%n)
          No intercept term and no weights
!
          cs%mean = 'Z'
          cs%weight = 'U'
          Allocate memory required by the regression routine
          cs%ldq = cs%n
          Allocate (cs%b(m),cs%cov((m*m+m)/2),cs%h(cs%n),cs%p(m*(m+ & 
            2)),cs%q(cs%ldq,m+1),cs%res(cs%n),cs%se(m),cs%wk(m*m+5*(m- &
            1)),cs%wt(1),cs%isx(m))
        End If
        Set up the initial feature set.
        If DROP = 0, this is the Null set (i.e. no features).
1
        If DROP = 1 then this is the full set (i.e. all features)
        cs%isx(1:m) = drop
        Add (if DROP = 0) or remove (if DROP = 1) the all the features specified
1
        in Z
        inv\_drop = abs(drop-1)
        Do i = 1, 1z
          cs%isx(z(i)) = inv_drop
```

```
End Do
        Do i = 1, max(la,1)
          If (la>0) Then
            If (i>1) Then
              Reset the feature altered at the last iteration
              cs%isx(a(i-1)) = drop
            End If
!
            Add or drop the I'th feature in A
            cs%isx(a(i)) = inv_drop
          End If
          ip = count(cs%isx(1:m)==1)
          Fit the regression model
!
          ifail = 0
          Call g02daf(cs%mean,cs%weight,cs%n,cs%x,cs%ldx,m,cs%isx,ip,cs%y, &
            cs%wt,rss,idf,cs%b,cs%se,cs%cov,cs%res,cs%h,cs%q,cs%ldq,svd,irank, &
            cs%p,cs%tol,cs%wk,ifail)
          Return the score (the residual sums of squares)
         bscore(i) = rss
        End Do
      End Subroutine calc_subset_score
      Subroutine free_subset_score(cs)
        .. Scalar Arguments ..
       Type (calc_subset_data), Intent (Inout) :: cs
!
        .. Intrinsic Procedures ..
        Intrinsic
                                             :: allocated
        .. Executable Statements ..
!
        If (allocated(cs%x)) Then
          Deallocate (cs%x)
        End If
        If (allocated(cs%y)) Then
          Deallocate (cs%y)
        End If
        If (allocated(cs%b)) Then
         Deallocate (cs%b)
        End If
        If (allocated(cs%cov)) Then
          Deallocate (cs%cov)
        End If
        If (allocated(cs%h)) Then
          Deallocate (cs%h)
        End If
        If (allocated(cs%p)) Then
         Deallocate (cs%p)
        End If
        If (allocated(cs%q)) Then
          Deallocate (cs%q)
        End If
        If (allocated(cs%res)) Then
          Deallocate (cs%res)
        End If
        If (allocated(cs%se)) Then
         Deallocate (cs%se)
        End If
        If (allocated(cs%wk)) Then
          Deallocate (cs%wk)
        End If
        If (allocated(cs%wt)) Then
          Deallocate (cs%wt)
        End If
        If (allocated(cs%isx)) Then
          Deallocate (cs%isx)
        End If
     End Subroutine free_subset_score
    End Module h05aafe_mod
```

H05AAF.10 Mark 24

```
Program h05aafe
      .. Use Statements ..
!
      Use nag_library, Only: h05aaf, nag_wp
      Use h05aafe_mod, Only: calc_subset_data, calc_subset_score,
                                                                                 &
                             free_subset_score, nin, nout
      .. Implicit None Statement ..
1
      Implicit None
      .. Local Scalars ..
      Type (calc_subset_data)
                                           :: cs
     Real (Kind=nag_wp)
                                            :: gamma
     Integer
                                            :: cnt, drop, i, ifail, ip, irevcm, &
                                               j, la, licomm, lrcomm, lz, m, mincnt, mincr, mip, nbest
      .. Local Arrays ..
     Real (Kind=nag_wp)
                                            :: acc(2)
      Real (Kind=nag_wp), Allocatable
                                           :: bscore(:), rcomm(:)
      Integer, Allocatable
                                            :: a(:), bz(:,:), ibz(:), icomm(:), &
                                               z(:)
     Logical, Allocatable
                                            :: mask(:)
      .. Intrinsic Procedures ..
!
      Intrinsic
                                            :: max, pack
!
      .. Executable Statements ..
      Write (nout,*) 'HO5AAF Example Program Results'
     Write (nout,*)
      Skip headings in data file
!
     Read (nin,*)
     Read (nin,*)
      Read in the problem size
     Read (nin,*) m, ip, nbest
     Read in the control parameters for the subset selection
      Read (nin,*) mincr, mincnt, gamma, acc(1:2)
     Allocate memory required by the subset selection routine
     mip = m - ip
     Allocate (z(mip),a(max(nbest,m)),bz(mip,nbest),bscore(max(nbest,m)))
     Allocate dummy communication arrays, as we will query the required size
!
      licomm = 2
      1rcomm = 0
      Allocate (icomm(licomm),rcomm(lrcomm))
!
      Query the required length for the communication arrays
      irevcm = 0
      ifail = 1
      Call hO5aaf(irevcm,mincr,m,ip,nbest,drop,lz,z,la,a,bscore,bz,mincnt, &
        gamma,acc,icomm,licomm,rcomm,lrcomm,ifail)
     Extract the required sizes from the communication array
!
      licomm = icomm(1)
      lrcomm = icomm(2)
     Reallocate communication arrays to the correct size
      Deallocate (icomm, rcomm)
      Allocate (icomm(licomm),rcomm(lrcomm))
      Initialise reverse communication control flag
!
      irevcm = 0
      Call the reverse communication best subset routine in a loop.
      The loop should terminate when IRECVM = 0 on exit
      cnt = 0
rev_lp: Do
        ifail = -1
        Call h05aaf(irevcm,mincr,m,ip,nbest,drop,lz,z,la,a,bscore,bz,mincnt, &
          gamma,acc,icomm,licomm,rcomm,lrcomm,ifail)
        If (irevcm==0) Then
          If (ifail==0 .Or. ifail==53) Then
```

```
No error, or a warning was issued
           Exit rev_lp
         Else
           An error occurred
           Go To 100
         End If
       End If
       Calculate and return the score for the required models and keep track
!
!
       of the number of subsets evaluated
       cnt = cnt + max(1,la)
       Call calc_subset_score(m,drop,lz,z,la,a,bscore,cs)
     End Do rev_lp
     Call free_subset_score(cs)
1
     Titles
     Write (nout,99999) '
                            Score
                                        Feature Subset'
     Write (nout,99999) '
                             ----
                                          -----'
     Display the best subsets and corresponding scores. HO5AAF returns a list
     of features excluded from the best subsets, so this is inverted to give
!
     the set of features included in each subset
     Allocate (ibz(m), mask(m))
      ibz(1:m) = (/(i,i=1,m)/)
     Do i = 1, la
       mask(1:m) = .True.
       Do j = 1, mip
         mask(bz(j,i)) = .False.
       End Do
       Write (nout,99998) bscore(i), pack(ibz,mask)
     End Do
     Write (nout,*)
     If (ifail==53) Then
       Write (nout,99997) nbest, &
          ' subsets of the requested size do not exist, only ', la, &
         ' are displayed.'
     End If
     Write (nout, 99996) cnt, ' subsets evaluated in total'
100
    Continue
99999 Format (1X,A)
99998 Format (1X,E12.5,100(1X,I5))
99997 Format (1X,IO,A,IO,A)
99996 Format (1X, IO, A)
   End Program h05aafe
9.2
    Program Data
HO5AAF Example Program Data
Data required by HO5AAF
                          :: M, IP, NBEST
0 -1 -1.0 -1.0 -1.0
                          :: MINCR, MINCNT, GAMMA, ACC(1:2)
Data required by the scoring function
                          :: N
1e-6
                          :: TOL
-1.59
       0.19
              0.40
                     0.43 -0.40
                                  0.79
                                         0.06
 0.33
        1.60
              0.58 -1.12 1.23
                                  1.07 -0.07
                                                          -2.44
                            0.61
                                  -2.05
                                         -0.02
        0.61
              -0.36
                     1.16
-0.25
-0.04
        0.80
              -0.73
                     -0.63
                            -0.75
                                  -0.73
                                          1.43
                                                          -2.97
-2.28
                     0.33 0.16
                                  -0.21 -1.61
        0.46 -0.65
-0.54
        0.48
              0.37 -0.95 -2.14
                                  0.48 2.02
                                                          7.42
                                  0.23
        1.05
              0.64 0.02 -1.12
                                         0.06
-0.52
        1.40
              -0.98
                      2.47
                            0.49
                                  -0.02
                                                           3.00
 -1.26
                                         -0.05
                     0.73 -1.41
                                         0.20
        1.86
              0.10
                                  0.98
-0.84
```

H05AAF.12 Mark 24

-0.89	1.84	2.56	0.60	-0.12	0.71	0.23	8.83
1.12	-0.51 -1.04	-0.58	0.09 -0.01	-1.14	2.11	-0.11 0.05	
0.06	0.85 0.80	-2.09	0.22 0.18	<b>-1.</b> 35	-0.36	1.20 0.63	2.57
	-1.02 0.45	0.08 0.62	-0.06 -1.97	0.13 0.97	-1.18 0.93	2.30 -0.18	
0.08 0.65	-0.31	0.43	-0.38	0.01	1.30	0.66 -0.90	4.55
	1.14 -1.00	0.76 0.40 -0.13	2.37 -0.69	1.10 -0.62		-0.38 0.00	
	-0.21 1.26	-0.51			0.29	0.38 0.68	3.38
	-1.59 1.05	-0.58 1.30	-1.09 -0.98			-1.02 -1.32	
	-1.58 -0.73		2.19	-0.49	0.69		5.47
-0.63	-3.09 1.20	-0.61 -0.04	-1.89 1.02	0.15 0.31	0.77 0.81	-0.49 -0.45	13.97
2.24		0.30	1.39	-0.38	-0.71	0.24 0.48	20.94
0.36	-0.26	-1.40 0.08	0.06	-1.49	0.43		<b>-</b> 12.87
-1.33	1.50 -0.96	-0.02	-0.25 0.51	-1.38	-0.43 -0.78	1.90 1.82	28.26
-0.57	-2.69	-0.64	-0.34			0.61 -0.19	6.89
0.90	-2.20	-0.38 1.72	0.29	0.66	0.19	-0.57	5.37
0.36		-0.41 1.87	1.22 -1.48	-0.30 0.52	0.77 1.35	0.82	-1.50
-0.40 -0.34	-1.10 -0.70	-0.83	0.71	0.27	0.37	1.30 -1.79 0.87	-23.00
1.01	1.20	0.28	0.79	2.76	0.35	1.31	14.09
1.09		-0.59 -0.34	-0.03	-0.59	-1.70	1.31 -0.03	-11.05
-1.36	-1.45 0.13	0.70 0.70	2.10 -1.51	0.08	-0.62	-0.38 -0.64	-32.04
<b>-1.</b> 58	0.00	0.58	-0.21	1.30	2.02	-0.14 1.52	23.36
						0.91 -1.74	
-1.92		0.56		-0.71	1.25	-0.72 0.10	
-0.01	0.83		-0.78	0.50	-1.22	0.54	-5.30
	0.46	0.15	-1.96 1.58	0.93		-0.23 -0.26	-7.77
-0.91 -1.39 0.65	0.23 -0.30 -1.14	-0.81		-0.27 0.88 -0.68	-0.09	-0.60 -0.35 0.21	-34.25
0.94 -0.36		0.81	-0.33	0.42	-0.90	0.49 -1.35	26.78
-0.64	1.20		0.18	-1.39 0.62	-0.81	-0.99 -1.08	-11.85
<b>-</b> 2 15	1.37		-1.48	-0.79	0.28	-0.20 0.62	-8.62
	0.39	1.14	0.15	1.03	0.46	0.40	12.35
-1.77 -0.48	0.72	-2.05	-0.03 0.27	-1.24 -0.84	-1.40	-0.06 -3.67	-1.54
0.09 -1.65		-0.30			0.00	0.43	-16.59
-1.72 -0.85	1.08 -0.02	-0.94	0.49 -1.16	-0.56 0.49	0.95 1.56	1.09 -0.60	-8.69
0.32	0.72 -0.73	-1.20	2.52 1.50	1.78	0.16	-0.01 -0.65	7.82
0.68		0.40		-2.10		-0.18	-18.56
		1.14		-0.92		1.58	17.21 :: X,Y

# 9.3 Program Results

HO5AAF Example Program Results

Score	Feat	Feature Subset							
0.10475E+04	4	7	8	10	14				
0.10599E+04	4	5	7	8	14				
0.10702E+04	4	5	7	10	14				

45 subsets evaluated in total

H05AAF.14 (last) Mark 24